# $Id$
# Maintainer: Felix Yan <felixonmars@gmail.com>

pkgname=perl-file-copy-recursive
pkgver=0.38
pkgrel=3
pkgdesc="Perl extension for recursively copying files and directories"
arch=('any')
url="https://metacpan.org/release/File-Copy-Recursive"
license=('GPL' 'PerlArtistic')
depends=('perl')
options=('!emptydirs')
source=("http://cpan.metacpan.org/authors/id/D/DM/DMUEY/File-Copy-Recursive-${pkgver}.tar.gz")
sha512sums=('49e5dbfc5bdbc554ad8a019c25db85ebfa55765a73758d95271b5f3de9550fd68b4d99aa287898d86bc6c11ea8de97469a85c53d4ad8534594b411be1f18eb87')

build() {
  cd File-Copy-Recursive-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check () {
  cd File-Copy-Recursive-$pkgver
  make test
}

package () {
  cd File-Copy-Recursive-$pkgver
  make DESTDIR="${pkgdir}" install
}

